
Система направлений и ее применение в торговле

Торговые системы на основе трендовых индикаторов

Считается, что системы, ориентированные на отслеживание трендов, в среднем дают большую прибыль, чем разворотные системы. 

В период консолидации рынка трендовая система может принести серию убытков: она с запаздыванием дает сигналы и позиции, открытые по ним, не успевают принести прибыль на рынке, быстро изменяющем свое направление. Однако даже небольшое число удачно открытых позиций (в начале длительных трендов) могут не только компенсировать эти убытки, но и принести высокую прибыль.

Трендовые системы имеют явное преимущество в том, что они обычно дают меньшее число торговых сигналов. Чем меньше времени трейдер имеет открытую позицию, тем меньше вероятность получить убыток при внезапном изменении рыночной ситуации. Кроме того, большое число открываемых позиций связано с большими расходами на комиссионные и потери при открытии позиций (проскальзывания при исполнении ордеров и т.д.).

Эти рассуждения заслуживают внимания и со стороны индивидуальных валютных трейдеров, но в еще большей степени они требуют тщательной проверки. Многие общие рекомендации, известные по учебникам технического анализа и торговым системам, оказываются малоприменимыми на часовых графиках валют, где волей-неволей происходит основная деятельность валютного трейдера. Длительных направленных ходов на графиках валют, при внимательном рассмотрении, оказывается не так уж много. И большинство из них выглядят как убедительные тренды только задним числом, в реальной же торговле в любой момент времени приходится решать одну и туже проблему: является ли данное движение цены откатом или же началом нового тренда. Однозначного ответа не может дать ни один подход! 

Поэтому, открывая каждую позицию, для нее надо предусматривать достаточно жесткий стоп, ибо каждое движение против позиции может превратиться в противоположный ход рынка. Следует намечать ближнюю цель для фиксирования прибыли и только потом уже искать аргументы в пользу длительного сопровождения этой позиции в случае удачного хода. Одним словом, реальная торговля, независимо от того, на каких правилах или индикаторах она построена, носит все признаки именно краткосрочного торгового подхода, ориентированного на получение небольших прибылей и жесткое ограничение риска. Нет сомнения, что большинство трейдеров следуют именно такому стилю торговли, просто в силу его большей психологической доступности в условиях ограниченного размера капитала.

Тем не менее, введение в торговые планы элементов трендовых систем чрезвычайно полезно и необходимо. Большие прибыли, хотя и случаются редко (в литературе упоминаются оценки, что менее 5% всех позиций приносят основную часть зарабатываемой прибыли), но упускать саму возможность их получения нельзя. Поэтому в арсенале трейдера должны быть готовые к применению трендовые торговые системы, а в процессе торговли следует постоянно искать в поведении рынков признаки, позволяющие принимать уверенные решения о необходимости следования длительной тенденции.

Возможных вариантов построения трендовых торговых систем очень много. Здесь будет рассмотрен подход, основанный на использовании популярного индикатора Система Направлений. 
Система направлений (Directional Movement System, DMS) – набор индикаторов, хорошо решающий задачу обнаружения тренда на заданном рынке. 

Вычисление. DMS (иногда обозначаемый также DMI - Directional Movement Index) включает три отдельных индикатора, вычисляемых на основе анализа диапазона (размаха) изменения цен. Система направлений является формализацией исходного понятия тенденции, принятого в техническом анализе: рынок растет, когда на графике появляется серия из последовательно сдвигающихся вверх размахов рынка (более высокие максимумы и более высокие минимумы); рынок падает, когда график демонстрирует серию последовательно сдвигающихся вниз размахов рынка (понижающиеся максимумы и понижающиеся минимумы). 

В первом случае диапазон (от low до high) последовательных свечей имеет склонность сдвигаться все выше, во втором – все ниже. Основными количественными мерами для оценки таких сдвигов торгового диапазона являются два показателя: +DM – показатель степени движения рынка вверх, и –DM – показатель степени движения рынка вниз. По определению, +DM есть наибольшая из частей (верхняя или нижняя) диапазона последней свечи, находящихся за пределами диапазона предыдущей свечи.

Рисунок 1 поясняет процедуру вычисления этих величин.

Рисунок 1. Вычисление показателей направленности движения

Здесь изображены пять возможных конфигураций расположения двух свечей: последней свечи, для которой и вычисляются DM, и предыдущей:
A) диапазон последней свечи выходит вверх за пределы диапазона предыдущей, величина, на которую последний диапазон превышает предыдущий, берется в качестве +DM, а -DM = 0;

B) диапазон последней свечи выходит вниз за пределы диапазона предыдущей, величина, на которую последний диапазон сдвинут ниже предыдущего, берется в качестве -DM, а +DM = 0;

C) если диапазон последней свечи целиком расположен внутри диапазона предыдущей, то направленное движение отсутствует, оба показателя равны нулю;

D) диапазон последней свечи выходит и вниз и вверх  за пределы диапазона предыдущей, но величина сдвига вниз больше, чем величина сдвига вверх, она и берется в качестве -DM, а +DM = 0;

E) диапазон последней свечи выходит и вниз и вверх  за пределы диапазона предыдущей, но величина сдвига вверх больше, чем величина сдвига вниз, она и берется в качестве +DM, а -DM = 0.

Следующий шаг – вычисление показателя “истинный диапазон” (True Range, TR), который определяется как наибольшее из трех чисел,

TR = max {D1, D2, D3},

где  D1 – диапазон последней свечи (расстояние от high до low для последней свечи);

       D2 – расстояние от закрытия предыдущей свечи (close на Рисунке 2) до максимума последней;

       D3 – расстояние от закрытия предыдущей свечи до минимума последней.

Рисунок 2. К вычислению показателя True Range

Затем вычисляются показатели направления (Directional Index или Directional Indicator),

+DI = MA(+DM, N) / MA(TR, N)

-DI = MA(-DM, N) / MA(TR, N)

представляющие собой сглаженные, нормированные по отношению к TR, показатели направленности движения +DM. Смысл нормировки заключается в том, что с помощью таких показателей можно сравнивать разные рынки, так как в отличие от +DM, измеряемых в пунктах, +DI измеряются в процентах. Сглаживание с помощью скользящего среднего показатели устраняет слишком резкие колебания показателей, выделяя основные изменения.

+DI – первые две линии, выводимые на графике DMS (Рисунок 3); +DI растет на бычьем тренде, а –DI растет на медвежьем тренде. Длина N окна сглаживания скользящего среднего является основным параметром системы направлений.

Третьим индикатором системы DMS является ADX, 

ADX = 100*MA( (+DI – (-DI)( / (+DI + (-DI)), N),

который показывает (в процентном измерении), насколько одно из направлений движения рынка преобладает над другим. Информация о том, какое именно направление – рост или снижение – является господствующим, в индикаторе ADX уже не содержится, он принимает большие значения в том случае, когда на рынке есть достаточно явно выраженная тенденция (определяемая на основе показателей +DI), но неважно, какая именно, бычья или медвежья.

Сглаженный с периодом N показатель True Range называется Average True Range (ATR) и часто используется в торговых решениях как мера волатильности цены.

Рисунок 3. Система направлений на часовом графике швейцарского 

франка (параметр усреднения N = 14):

красная линия –DI, синяя линия +DI, зеленая линия ADX

Интерпретация. Система направлений широко применяется в различных торговых подходах для принятия решений о наличии на рынке достаточно явно выраженной тенденции изменения цен. 

Основные правила интерпретации системы направлений:

· когда +DI находится выше –DI, на рынке бычья тенденция;

· когда -DI находится выше +DI, на рынке медвежья тенденция;

· показатель ADX растет, когда расстояние между линиями +DI увеличивается, то есть, одна тенденция явно преобладает над другой, а при сближении этих линий ADX падает (имевшая место тенденция выдыхается); 

· если ADX находится ниже обеих линий направления +DI, то это признак консолидации на рынке, после нее начнется новый ход в сторону той тенденции, которая победит.

На этом понимании системы направлений основаны стандартные рекомендации по ее использованию в торговле:

· buy когда +DI находится выше –DI и ADX растет;

· sell когда -DI находится выше +DI и ADX растет;

· когда ADX начинает расти, находясь ниже обеих линий направления, это удобный момент для открытия позиции (в сторону верхней из линий направлений), когда же ADX, находящийся высоко над обеими линиями направления, начинает идти вниз – самое время для фиксирования прибыли в позициях, открытых ранее в сторону господствующей тенденции.

Параметры. Стандартная рекомендация для параметра системы направлений N = 14. Но эта рекомендация пришла с дневных графиков фондового рынка; на внутридневных масштабах графиков валют требуется достаточно тщательный анализ для  обоснованного выбора параметра.

Применение в торговых системах. Главное свойство системы направлений DMS, на котором строятся ее применения в торговле: DMS - надежный детектор тренда, хотя часто и запаздывающий. Поэтому в современной практике DMS обычно используют в системах именно как указатель тренда, а сами решения об открытии позиций принимается на основе других индикаторов.

Но создатель системы направлений Welles Wilder в своей книге New Concepts in Technical Trading Systems рассматривал способ применения DMS, основанный на одной лишь DMS; для этого он ввел понятие “экстремальной точки”. Экстремальная точка (Extreme Point, ET) есть один из экстремумов свечи, на которой произошло пересечение линий +DI: если +DI пересекает –DI снизу вверх, то в качестве ЕТ принимается high, если же -DI пересекает +DI снизу вверх, то в качестве ЕТ принимается low. Экстремальная точка является ориентиром для открытия соответствующей позиции: после пересечения +DI линии –DI снизу вверх длинная позиция открывается, когда цена поднимается выше ЕТ; соответственно, после пересечения -DI линии +DI снизу вверх короткая позиция открывается, когда цена опускается ниже ЕТ.

В системе ADX_Wilders вход в позицию осуществляется на пересечении +DI и -DI, при условии, что ADX в момент пересечения поднимается; уровень открытия позиции – экстремум свечи, на которой имело место пересечение +DI и –DI (экстремальные точки Wilder’a Wilders_max и Wilders_min):

       Wilders_max = (ValueWhen(Cross(PDI,MDI) AND ADX[1] < ADX), LOW)

       Wilders_min = (ValueWhen(Cross(MDI,PDI) AND ADX[1] < ADX), HIGH)

                  Enter Long  Cross (C, Wilders_max)

                  Enter Short Cross (Wilders_min, C)

Применение этой системы показывает, что такие правила являются слишком жесткими, они ограничивают возможности системы по открытию позиций, пропуская тем самым начала многих хороших ходов. 

Все-таки по своему основному назначению DMS является дополнительным индикатором, который должен использоваться вместе с некоторыми другими индикаторами и правилами, определяющими моменты открытия позиций. При этом DMS служит для выбора наиболее надежных сигналов: если DMS показывает наличие восходящей тенденции, то выбираются только buy сигналы среди тех, которые генерируются другими индикаторами и правилами, если же DMS показывает медвежью тенденцию, то выполняются только sell сигналы. Наличие тенденции определяется, с точки зрения DMS, следующим критерием: когда ADX больше некоторого порогового значения (обычно принимается около 25), то на рынке имеет место тенденция, направленная в ту сторону, чья линия, -DI или +DI находится в это время выше другой.

Например, в системе ADX&MA ослаблено условие открытия позиции по пересечению +DI и –DI: индикатор ADX в этот момент не обязательно должен расти (то есть позиция по пересечению +DI и –DI открывается и на нетрендовом рынке); но в систему добавлена возможность открытия позиции по пересечению цены со скользящей средней, в этом случае ADX должен быть выше некоторого уровня (то есть позиция по пересечению с МА открывается только на трендовом рынке). В таком виде система обещает быть более универсальной, она может пройти с прибылью и участки рынка с сильными откатами цены (Рисунок 4).

Рисунок 4. Система ADX&MA на часовом графике швейцарского франка

В целом, используя систему направлений совместно с другими индикаторами, следует иметь в виду, что растущий ADX означает наличие тренда, так что в этом случае лучше пользоваться трендовыми системами; падающий ADX означает переход рынка в консолидацию, здесь лучше должны работать разворотные системы на основе осцилляторов.

Вообще говоря, возможны различные понимания тренда, определяемого в соответствии с ADX:

А) имеет место тренд, когда ADX выше некоторого критического уровня (обозначим этот уровень levelADX, часто упоминаемые в литературе значения levelADX равны 20 или 24), при ADX < levelADX рынок считается нетрендовым (разумеется, конкретное значение такого критического уровня зависит и от самого рынка и от выбранного временного параметра индикатора ADX, стандартная рекомендация периода усреднения 14, но на внутридневных графиках валют все это требует выполнения специальных анализов);

Б) имеет место тренд, когда ADX > levelADX, либо ADX растет (хотя при этом может быть и ADX < levelADX), направление тренда в обоих случаях указывается взаимным расположением индикаторов +DI.

Заслуживает отдельного исследования вопрос: что имеет более важное значение - уровень ADX (в частности, считается, что разворот вниз ADX, поднявшегося перед этим выше уровня 40, означает переход рынка в консолидацию) или темп изменения ADX? И как ответ на этот вопрос зависит от того, какой индикатор применяется для открытия позиций? 

Полезным ориентиром на графике ADX является локальный максимум. Если ADX, находящийся достаточно высоко, показывает разворот вниз, то это вполне может означать, что предшествующий энергичный тренд (поскольку ADX высоко, то перед этим обязательно был тренд!) закончился и начался разворот в другую сторону, либо же имеет место откат, после которого продолжится предшествующая тенденция. Одним из способов использования такого сценария является открытие позиции на откате. 

Предположим, что имел место бычий тренд (+DI > -DI), в результате которого ADX поднялся до высокого уровня (например > 30), затем цена сделала откат вниз. В этом месте ADX покажет локальный максимум и начнет снижаться. Но если предшествующая тенденция достаточно сильна, то ход вверх продолжится: цена, отразившись от некоторого уровня поддержки, пойдет вверх. На этом откате от уровня поддержки и следует открывать длинную позицию.

Такая идея реализована в системе ADXtop&Pullback , где позиции открываются после появления максимума ADX и отката цены до границы канала. Условие локального максимума на графике ADX в ней формулируется следующим образом:

ADX[2] < ADX[1] AND ADX[1] > ADX,

где ADX – текущее значение индикатора, ADX[1] – значение на предыдущей свече, ADX[2] – значение две свечи назад. Условием состоявшегося отката является прорыв ценового канала; для открытия длинной позиции цена должна закрыться ниже уровня поддержки канала LLV(Ref(LOW, -1), lookback) (lookback – длина окна просмотра канала отката), для открытия короткой позиции цена должна закрыться выше уровня сопротивления канала HHV(Ref(HIGH, -1), lookback).

Открытие позиции после подъема и формирования локального максимума ADX возможно также использовать как антитрендовый метод, поскольку поднимающийся ADX является результатом присутствия длительного тренда, который рано или поздно должен переломиться.

Некоторые другие модификации, заслуживающие исследования:

А) Использовать разные периоды в +DI и –DI для открытия длинных и коротких позиций.

Б) Применять ADX меньшей длины при высоких значениях ADX, чтобы индикатор успевал среагировать в момент обратного разворота рынка.

Две следующие системы, предложенные в книге T.Chande [2], демонстрируют разнообразие возможных вариантов применения ADX в торговых решениях. Согласно наблюдениям за поведением многих рынков, скачок величины ADX часто предшествует продолжительному развитию тренда. Поэтому, обнаружив такой импульс (│ADX – ADX[1]│> 1), имеет смысл открывать позицию в направлении, указываемом каким-либо трендовым индикатором, например, расположением двух скользящих средних (ADX-Burst-2МА System – импульсная трендовая система). Разумеется, необходимо провести специальное тестирование, чтобы убедиться в применимости этой системы для внутридневных графиков валютного рынка.
Система TAT System (Trend-antitrend system) объединяет в себе два варианта торговли в ценовом канале [1]: на трендовом рынке она открывает позицию при прорыве границы канала, а на нетрендовом рынке открывает позиции в направлении отката цены внутрь канала. Система как бы пытается сначала открыть позицию против тренда, а потом ищет повторный вход в рынок по тренду.

Границы канала (channel) понимаются в ней как линии поддержки и сопротивления LowestLow и HighestHigh, а критерием наличия тренда является расположение ADX и его скользящей средней: при ADX > MA(ADX) рынок считается трендовым (ADX растет), а при ADX < MA(ADX) рынок рассматривается как нетрендовый (поскольку ADX падает). 

На подобных идеях можно построить множество торговых систем, используя разнообразные торговые сигналы (например, от осцилляторов, как в приведенной ниже системе Stochastic_os/ob&DMItrend), варианты определения ценовых диапазонов и различные критерии тренда, основанные на DMS.

Приложение

Тексты торговых систем с использованием DMS
ADX_Wilders

Открытие позиции на пересечении +DI и -DI, при условии, что ADX растет. Уровень открытия позиции – экстремум свечи, на которой имело место пересечение +DI и –DI. Этот экстремум должен быть достигнут не позже, чем на opt2 свече после пересечения +DI и –DI.

Enter Long

Cross(C, ValueWhen(1, Cross(PDI(opt1),MDI(opt1)) AND Ref(ADX(opt1),-1)< ADX(opt1), HIGH) ) AND

BarsSince(Cross(PDI(opt1),MDI(opt1)) AND Ref(ADX(opt1),-1) < ADX(opt1)) < opt2

Enter Short

Cross(ValueWhen(1, Cross(MDI(opt1),PDI(opt1)) AND Ref(ADX(opt1),-1) < ADX(opt1), LOW),C) AND

BarsSince(Cross(MDI(opt1),PDI(opt1)) AND Ref(ADX(opt1),-1) < ADX(opt1)) < opt2

ADX&MA
Открытие позиции в начале нового движения (при пересечении +DI и –DI), либо на трендовом рынке (ADX выше порога opt4) при пересечении цены и скользящей средней

Enter Long

Cross(C,ValueWhen(1,Cross(PDI(opt1),MDI(opt1)),HIGH)) AND

BarsSince(Cross(PDI(opt1),MDI(opt1))) < opt2

OR C > Mov(C, opt3, S) AND ADX(opt1) > opt4 
AND PDI(opt1)>MDI(opt1)
Enter Short

Cross(ValueWhen(1,Cross(MDI(opt1),PDI(opt1)), LOW),C) AND

BarsSince(Cross(MDI(opt1),PDI(opt1))) < opt2

OR C < Mov(C, opt3, S) AND ADX(opt1) > opt4

AND MDI(opt1)>PDI(opt1)

ADXtop&Pullback
Открытие позиций на откате от границы канала после появления максимума ADX

Enter Long

BarsSince(Ref(ADX(opt1),-2) <= Ref(ADX(opt1),-1) AND

Ref(ADX(opt1),-1) > ADX(opt1) AND ADX(opt1) > opt2 

AND  PDI(opt1) >  MDI(opt1) ) < opt3 AND 

C < LLV(Ref(LOW, -1), opt4)

Enter Short

BarsSince(Ref(ADX(opt1),-2) <= Ref(ADX(opt1),-1) AND

Ref(ADX(opt1),-1) > ADX(opt1) AND ADX(opt1) > opt2 

AND  PDI(opt1) <  MDI(opt1) ) < opt3 

AND C > HHV(Ref(HIGH, -1), opt4)

ADX-Burst-2МА System

ADX-импульсная трендовая система

Enter Long

Mov(C, opt1, S) > Mov(C, opt2, S) AND Abs(ADX(opt3)- Ref(ADX(opt3),-1)) > 1

Enter Short

Mov(C, opt1, S) < Mov(C, opt2, S) AND Abs(ADX(opt3)- Ref(ADX(opt3),-1)) > 1

Замечание. Вспомогательный индикатор, полезный при компьютерном исследовании системы: ADXburst - величина 1-периодного изменения индикатора ADX, т.е. Momentum1(ADX)

Periods:= Input("Enter ADX periods", 8, 144, 14);

Abs(ADX(Periods)- Ref(ADX(Periods),-1))

Индикатор ADXburst определяет момент открытия позиции (ADXburst>1 указывает на скачок графика), а направление позиции диктуется расположением двух МА).

TAT System

Trend-antitrend system.

При растущем ADX (ADX>MA(ADX)) open на выходе из канала (buy new high, sell new low), при падающем ADX open внутрь канала ((buy new low, sell new high)

Enter Long

H > HHV(Ref(HIGH,-1),opt1) AND ADX(opt2) > Mov(ADX(opt2), opt3, S) OR  

L < LLV(Ref(LOW, -1), opt1) AND ADX(opt2) < Mov(ADX(opt2), opt3, S)

Enter Short

H > HHV(Ref(HIGH,-1),opt1) AND ADX(opt2) < Mov(ADX(opt2), opt3, S) OR  

L < LLV(Ref(LOW, -1), opt1) AND ADX(opt2) > Mov(ADX(opt2), opt3, S)

Stochastic_os/ob&DMItrend
Открытие позиций при пересечении осциллятором Stochastic уровней перекупленности/перепроданности, а также при пресечении быстрой и медленной линий Stochastic в направлении, указываемом DMS

Enter Long

Cross(Stoch(opt1,opt2),opt3) OR 

Cross(Stoch(opt1,opt2),Mov(Stoch(opt1,opt2), opt5,S)) AND

PDI(opt6) > MDI(opt6) AND ADX(opt6)> Ref(ADX(opt6), -1) 

Enter Short

Cross(opt4,Stoch(opt1,opt2))OR 

Cross(Mov(Stoch(opt1,opt2), opt5,S), Stoch(opt1,opt2)) AND

PDI(opt6) < MDI(opt6) AND ADX(opt6)> Ref(ADX(opt6), -1)







